Занятие 7 «Решение дифференциальных уравнений». «Моделирование динамических задач и систем. Расчет траектории камня с учетом сопротивления воздуха»

Для решения простых дифференциальных уравнений (задача Коши) используется функция dsolve

dsolve(ODE)
dsolve(ODE,y(x),extra_args)
dsolve({ODE,ICs},y(x),extra_args)
dsolve({sysODE,ICs},{funcs},extra_args)

Здесь ODE – обыкновенное дифференциальное уравнение, y(x)- функция одной переменной, ICs – выражение, задающее начальные условия, {sysODE}- множество дифференциальных уравнений, {funcs}- множество неопределенных функций, extra_argument – опция, задающая тип решения

- exact – аналитическое решение (принято по умолчанию);
- explicit – решение в явном виде;
- system – решение системы дифференциальных уравнений;
- ICs – решение системы дифференциальных уравнений с заданными начальными условиями;
- formal series – решение в форме степенного многочлена;
- series – решение в виде ряда с порядком, указываемым значением переменной Order;
- numeric –решение в численном виде.

Пример:

```
> dsolve(diff(y(x),x)*sqrt(1-x^2)=1+(y(x))^2,y(x));
```

$$y(x) = \tan(\arcsin(x) + \_C1)$$

Если требуется получить не общее решение, а общий интеграл обыкновенного дифференциального уравнения, то к аргументам функции dsolve() добавляется экстра-аргумент implicit.

```
> dsolve(diff(y(x),x)*sqrt(1-x^2)=1+(y(x))^2,y(x),implicit);
```

$$\arcsin(x) - \arctan(y(x)) + \_C1 = 0$$

Если необходимо найти частное решение дифференциального уравнения, отвечающее заданному начальному условию

$$y' tgx = y, \quad y\left(\frac{\pi}{2}\right) = 1.$$

```
> dsolve({diff(y(x),x)*tan(x)=y(x),y(Pi/2)=1},y(x));
```

$$y(x) = \sin(x)$$

Рассмотрим пример решения линейного неоднородного уравнения второго порядка с постоянными коэффициентами и специальной правой частью

$y'' + 5y' + 6y = e^{-x} + e^{-2x}$.

```
> dsolve(diff(y(x),x$2)+5*diff(y(x),x)+6*y(x)=exp(-x)+exp(-2*x),y(x));
```

$$y(x) = e^{(-3x)}\_C2 + e^{(-2x)}\_C1 + \frac{1}{2}(e^{x} - 2 + 2x)e^{(-2x)}$$

Системы дифференциальных уравнений решаются аналогично.

$$\begin{cases} \dot{x} = x + y - \cos(t), \\ \dot{y} = -2x - y + \sin t + \cos t. \end{cases}$$

Решение следующее:

```
> dsolve({diff(x(t),t)=x(t)+y(t)-cos(t),diff(y(t),t)=-2*x(t)-y(t)+sin(t)+cos(t)},{x(t),y(t)});
```

$$\{ y(t) = \cos(t)\ \_C2 - \sin(t)\ \_C1 + \sin(t)\ t - \sin(t)\ \_C2 - \cos(t)\ \_C1 - \sin(t) + \cos(t)\ t + \cos(t),\ x(t) = \sin(t)\ \_C2 + \cos(t)\ \_C1 + \sin(t) - \cos(t)\ t \}$$

С помощью функции collect() приводим подобные слагаемые:

```
> collect(%,t):collect(%,_C1):collect(%,_C2);
```

$$\{ x(t) = \sin(t)\ \_C2 + \cos(t)\ \_C1 + \sin(t) - \cos(t)\ t,\ y(t) = (-\sin(t) + \cos(t))\ \_C2 + (-\cos(t) - \sin(t))\ \_C1 + (\sin(t) + \cos(t))\ t + \cos(t) - \sin(t) \}$$

Если необходимо найти частное решение системы дифференциальных уравнений, то в первые фигурные скобки добавляются начальные условия. Найдем решение последней системы при x(0)=1, y(0)=2

```
> dsolve({diff(x(t),t)=x(t)+y(t)-cos(t),diff(y(t),t)=-2*x(t)-y(t)+sin(t)+cos(t),x(0)=1,y(0)=2},{x(t),y(t)});
```

$$\{ y(t) = 2\cos(t) - 4\sin(t) + \sin(t)\ t + \cos(t)\ t,\ x(t) = 3\sin(t) + \cos(t) - \cos(t)\ t \}$$

«Моделирование динамических задач и систем. Расчет траектории камня с учетом сопротивления воздуха»

Модель должна позволять:

Вычислять положение в любой момент времени.

Исходные данные:

Масса камня, начальные координаты, начальная скорость и угол броска камня.

Гипотезы, принятые для модели:

- Камень будем считать материальной точкой массой m, положение которой совпадает с центром масс камня;
- Движение происходит в поле силы тяжести с постоянным ускорением свободного падения g и описывается уравнениями классической механики Ньютона;
- Расчеты будем проводить с учетом сопротивления воздуха по закону $F_{тр} = A*V, где A = 0.1 Н \cdot с/м.$

Согласно законам Ньютона имеем уравнения движения

Без учетопротивления воздуха а сопротивления воздуха $m\dfrac{d^2x}{dt^2} = 0, \quad m\dfrac{d^2y}{dt^2} = -mg,$

с учета сопротивления воздуха $m\dfrac{d^2x}{dt^2} = -A\dfrac{dx}{dt}, \quad m\dfrac{d^2y}{dt^2} = -mg - A\dfrac{dy}{dt}.$

Итак, пусть камни с массой 500 и 100 грамм брошены под углом 45 градусов к горизонту со скоростью $V_0 = 20 м/с$. Найдем их баллистические траектории, если сила $F_{тр} = -A*V, где A = 0.1 Н \cdot с/м.$ Сравним их с траекториями, получающейся без учета сопротивления воздуха.

Решение

```
> restart;
> with(plots):
> V0x:=V0*cos(alpha); V0y:=V0*sin(alpha);
```

$$V0x := V0\cos(\alpha)$$

$$V0y := V0 \sin(\alpha)$$

Поскольку мы рассматриваем два случая: камень массой 500 г. и камень массой 100 г. И для каждого случая проводим вычисления с учетом сопротивления воздуха и без него, то мы должны составить 4 системы дифференциальных уравнений (ДУ).

```
> sys1:=massa[1]*diff(x(t),t$2)=-A[1]*diff(x(t),t),massa[1]*diff(y(t),t$2)=-A[1]*diff(y(t),t)-massa[1]*g;
```

$$sys1 := massa_1 \left(\frac{d^2}{dt^2} x(t)\right) = -A_1 \left(\frac{d}{dt} x(t)\right), massa_1 \left(\frac{d^2}{dt^2} y(t)\right) = -A_1 \left(\frac{d}{dt} y(t)\right) - massa_1\, g$$

```
> sys2:=massa[1]*diff(x(t),t$2)=-A[2]*diff(x(t),t),massa[1]*diff(y(t),t$2)=-A[2]*diff(y(t),t)-massa[1]*g;
```

$$sys2 := massa_1 \left(\frac{d^2}{dt^2} x(t)\right) = -A_2 \left(\frac{d}{dt} x(t)\right), massa_1 \left(\frac{d^2}{dt^2} y(t)\right) = -A_2 \left(\frac{d}{dt} y(t)\right) - massa_1\, g$$

```
> sys3:=massa[2]*diff(x(t),t$2)=-A[1]*diff(x(t),t),massa[2]*diff(y(t),t$2)=-A[1]*diff(y(t),t)-massa[2]*g;
```

$$sys3 := massa_2 \left(\frac{d^2}{dt^2} x(t)\right) = -A_1 \left(\frac{d}{dt} x(t)\right), massa_2 \left(\frac{d^2}{dt^2} y(t)\right) = -A_1 \left(\frac{d}{dt} y(t)\right) - massa_2\, g$$

```
> sys4:=massa[2]*diff(x(t),t$2)=-A[2]*diff(x(t),t),massa[2]*diff(y(t),t$2)=-A[2]*diff(y(t),t)-massa[2]*g;
```

$$sys4 := massa_2 \left(\frac{d^2}{dt^2} x(t)\right) = -A_2 \left(\frac{d}{dt} x(t)\right), massa_2 \left(\frac{d^2}{dt^2} y(t)\right) = -A_2 \left(\frac{d}{dt} y(t)\right) - massa_2\, g$$

Зададим исходные данные для расчета

```
> V0:=20;massa:=[0.5,0.1];A:=[0.1,0];alpha:=Pi/4;g:=9.8;
```

$$V0 := 20$$

$$massa := [0.5, 0.1]$$

$$A := [0.1, 0]$$

$$\alpha := \frac{\pi}{4}$$

$$g := 9.8$$

```
> massa:= [.5, .1];
```

$$massa := [0.5, 0.1]$$

Решаем ДУ

```
>
p1:=dsolve({sys1,x(0)=0,D(x)(0)=V0x,y(0)=0,D(y)(0)=V0y},{y(t),x(t)},type=numeric,output=listprocedure);
```

$$p1 := \left[t = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}), x(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}), \frac{d}{dt} x(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}), y(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}), \frac{d}{dt} y(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}) \right]$$

```
> p2:=dsolve({sys2,x(0)=0,D(x)(0)=V0x,y(0)=0,D(y)(0)=V0y},{y(t),x(t)},type=numeric,output=listprocedure):
> p3:=dsolve({sys3,x(0)=0,D(x)(0)=V0x,y(0)=0,D(y)(0)=V0y},{y(t),x(t)},type=numeric,output=listprocedure):
> p4:=dsolve({sys4,x(0)=0,D(x)(0)=V0x,y(0)=0,D(y)(0)=V0y},{y(t),x(t)},type=numeric,output=listprocedure):
```

Создадим графические обьекты - результаты решения систем ДУ

```
> a1:=odeplot(p1,[x(t),y(t)],0..3,color=green,view=[0..50,0..15],thickness=2):
> a2:=odeplot(p2,[x(t),y(t)],0..3,color=red,view=[0..50,0..15],thickness=2):
> a3:=odeplot(p3,[x(t),y(t)],0..3,color=blue,view=[0..50,0..15],thickness=2):
> a4:=odeplot(p4,[x(t),y(t)],0..3,color=black,view=[0..50,0..15],thickness=2):
```

Построим графики траектории

```
> t:=textplot([[25,8,`A=0.1`],[35,9,`A=0`]],color=blue,font=[TIMES,ROMAN,12]);
```

t := PLOT(TEXT([25., 8.], "A=0.1"), TEXT([35., 9.], "A=0"),
FONT(*TIMES*, *ROMAN*, 12), COLOUR(*RGB*, 0., 0., 1.00000000))

```
> t1:=textplot([[17,3,`A=0.1`],[35,9,`A=0`]],color=blue,font=[TIMES,ROMAN,12]);
```

$t1$:= PLOT(TEXT([17., 3.], "A=0.1"), TEXT([35., 9.], "A=0"),
FONT(*TIMES*, *ROMAN*, 12), COLOUR(*RGB*, 0., 0., 1.00000000))

```
> display({a1,a2,t},title=`Траектория полета тела массой 500г`,labels=[x,y],labelfont=[TIMES,ROMAN,14]);
>
```



Построим график для второго случая

```
> display({a3,a4,t1},title=`Траектория полета тела массой
100г`,labels=[x,y],labelfont=[TIMES,ROMAN,14]);
```



Вывод.
Из приведенных расчетов и графиков видно, что при учете силы сопротивления воздуха дальность и высота полета сильно уменьшается по сравнению с полетом в вакууме, и эта разница зависит от массы тела.

Задания
1, Решите уравнения:

- $x\sqrt{1+y^2}\,dx + y\sqrt{1+x^2}\,dy = 0$;
- $\dfrac{xdy}{\sqrt{1-y^2}} + \dfrac{ydx}{\sqrt{1-x^2}} = 0;$
- $\sqrt{\dfrac{1+\cos 2x}{1+\sin y}} + y' = 0;$
- $y' = \dfrac{\cos y - \sin y - 1}{\cos x - \sin x + 1};$
- $y' + 2xy = xe^{-x^2}$;
- $y' - \dfrac{y}{x\ln x} = x\ln x, y(e) = e^2/2$

В качестве постоянной е рекомендуется использовать exp(1)/

2. Найдите общий интеграл уравнения:

- $(x^2 + 2xy)dx + xydy = 0;$
- $(2x + y + 1)dx + (x + 2y - 1)dy = 0;$
- $(x + y + 2)dx + (2x + 2y - 1)dy = 0;$
- $(x - y - 4)dy + (x + y - 2)dx = 0.$

3. Решите уравнения:

- $y''' = x \sin x;$
- $y''' \sin^4 x = \sin 2x;$
- $y'' = 2\sin x \cos^2 x - \sin^3 x;$
- $y'' - y' - 2y = 0;$
- $y'' - 6y' - 25y = \sin x + 3\cos x;$
- $y'' - 4y' + 3y = e^{5x}; y(0) = 3, y'(0) = 9$.

Замечание > `dsolve({ODE,y(0)=3,D(y)(0)=9},y(x));`

4. Решите системы дифференциальных уравнений:

- $\frac{dx}{dt} = 2x + y, \frac{dy}{dt} = x + 2y; x(0) = 1, y(0) = 3;$
- $\frac{dx}{dt} = e^{3t} - y, \frac{dy}{dt} = 2e^{3t} - x;;$

5. Провести расчеты и построить графики для камней массами 200г. и 700г. брошеными под углом 30 градусов к горизонту со скоростью $V_0 = 30 м / с$. Найти их траектории, если сила $F_{тр} = -A*V, где A = 0.1 Н \cdot с / м.$ Сравнить их с траекториями, получающейся без учета сопротивления воздуха.